C06PCF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

C06PCF calculates the discrete Fourier transform of a sequence of n complex data values (using complex data type).

2 Specification

```
SUBROUTINE C06PCF(DIRECT, X, N, WORK, IFAIL)
CHARACTER*1       DIRECT
INTEGER           N, IFAIL
complex           X(N), WORK(2*N+15)
```

3 Description

Given a sequence of n complex data values z_j, for $j = 0, 1, \ldots, n-1$, this routine calculates their (**forward** or **backward**) discrete Fourier transform defined by

$$\hat{z}_k = \frac{1}{\sqrt{n}} \sum_{j=0}^{n-1} z_j \times \exp\left(\pm i \frac{2\pi jk}{n}\right), \quad k = 0, 1, \ldots, n-1.$$

(Note the scale factor of $\frac{1}{\sqrt{n}}$ in this definition.) The minus sign is taken in the argument of the exponential within the summation when the forward transform is required, and the plus sign is taken when the backward transform is required. A call of the routine with DIRECT = 'F' followed by a call with DIRECT = 'B' will restore the original data.

The routine uses a variant of the fast Fourier transform (FFT) algorithm (Brigham [1]) known as the Stockham self-sorting algorithm, which is described in Temperton [2].

4 References

[1] Brigham E O (1973) *The Fast Fourier Transform* Prentice–Hall

[2] Temperton C (1983) Self-sorting mixed-radix fast Fourier transforms *J. Comput. Phys.* **52** 1–23

5 Parameters

1: DIRECT — CHARACTER*1 *Input*

On entry: if the **F**orward transform as defined in Section 3 is to be computed, then DIRECT must be set equal to 'F'. If the **B**ackward transform is to be computed then DIRECT must be set equal to 'B'.

Constraint: DIRECT = 'F' or 'B'.

2: X(N) — ***complex*** array *Input/Output*

On entry: if X is declared with bounds (0:N−1) in the (sub)program from which C06PCF is called, then X(j) must contain z_j, for $j = 0, 1, \ldots, n-1$.

On exit: the components of the discrete Fourier transform. If X is declared with bounds (0:N−1) in the (sub)program from which C06PCF is called, then for $0 \leq k \leq n-1$, $\hat{z}_k$ is contained in X(k).

3: N — INTEGER *Input*

On entry: the number of data values, n. The total number of prime factors of N, counting repetitions, must not exceed 30.

Constraint: N $\geq$ 1.

4: WORK(2∗N+15) — ***complex*** array *Workspace*

The workspace requirements as documented for this routine may be an overestimate in some implementations. For full details of the workspace required by this routine please refer to the Users' Note for your implementation.

On exit: the real part of WORK(1) contains the minimum workspace required for the current value of N with this implementation.

5: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1

On entry, N $<$ 1.

IFAIL = 2

On entry, DIRECT is not equal to one of 'F' or 'B'.

IFAIL = 3

On entry, N has more than 30 prime factors.

IFAIL = 4

An unexpected error has occurred in an internal call. Check all subroutine calls and array dimensions. Seek expert help.

7 Accuracy

Some indication of accuracy can be obtained by performing a subsequent inverse transform and comparing the results with the original sequence (in exact arithmetic they would be identical).

8 Further Comments

The time taken by the routine is approximately proportional to $n \times \log n$, but also depends on the factorization of n. The routine is somewhat faster than average if the only prime factors of n are 2, 3 or 5; and fastest of all if n is a power of 2.

9 Example

This program reads in a sequence of complex data values and prints their discrete Fourier transform (as computed by C06PCF with DIRECT set to 'F').

It then performs an inverse transform, using C06PCF with DIRECT set to 'B', and prints the sequence obtained alongside the original data values.

9.1 Program Text

```
*     C06PCF Example Program Text.
*     Mark 19 Release. NAG Copyright 1999.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=20)
*     .. Local Scalars ..
      INTEGER          IFAIL, J, N
*     .. Local Arrays ..
      complex          WORK(2*NMAX+15), X(0:NMAX-1), XX(0:NMAX-1)
*     .. External Subroutines ..
      EXTERNAL         C06PCF
*     .. Intrinsic Functions ..
      INTRINSIC        real, imag
*     .. Executable Statements ..
      WRITE (NOUT,*) 'C06PCF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
   20 CONTINUE
      READ (NIN,*,END=100) N
      IF (N.GT.1 .AND. N.LE.NMAX) THEN
         DO 40 J = 0, N - 1
            READ (NIN,*) X(J)
            XX(J) = X(J)
   40    CONTINUE
         IFAIL = 0
*
         CALL C06PCF('F',X,N,WORK,IFAIL)
*
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Components of discrete Fourier transform'
         WRITE (NOUT,*)
         WRITE (NOUT,*) '              Real      Imag'
         WRITE (NOUT,*)
         DO 60 J = 0, N - 1
            WRITE (NOUT,99999) J, real(X(J)), imag(X(J))
   60    CONTINUE
*
         CALL C06PCF('B',X,N,WORK,IFAIL)
*
         WRITE (NOUT,*)
         WRITE (NOUT,*)
     +     'Original sequence as restored by inverse transform'
         WRITE (NOUT,*)
         WRITE (NOUT,*)
     +     '                 Original                 Restored'
         WRITE (NOUT,*)
     +     '              Real      Imag           Real      Imag'
         WRITE (NOUT,*)
         DO 80 J = 0, N - 1
            WRITE (NOUT,99999) J, XX(J), X(J)
   80    CONTINUE
         GO TO 20
      ELSE
         WRITE (NOUT,*) 'Invalid value of N'
```

```
      END IF
  100 CONTINUE
      STOP
*
99999 FORMAT (1X,I5,2(:5X,'(',F8.5,',',F8.5,')'))
      END
```

9.2 Program Data

```
C06PCF Example Program Data
    7
  (0.34907,  -0.37168)
  (0.54890,  -0.35669)
  (0.74776,  -0.31175)
  (0.94459,  -0.23702)
  (1.13850,  -0.13274)
  (1.32850,   0.00074)
  (1.51370,   0.16298)
```

9.3 Program Results

```
 C06PCF Example Program Results

 Components of discrete Fourier transform

              Real      Imag

    0     ( 2.48361,-0.47100)
    1     (-0.55180, 0.49684)
    2     (-0.36711, 0.09756)
    3     (-0.28767,-0.05865)
    4     (-0.22506,-0.17477)
    5     (-0.14825,-0.30840)
    6     ( 0.01983,-0.56496)

 Original sequence as restored by inverse transform

                 Original                Restored
               Real      Imag          Real      Imag

    0     ( 0.34907,-0.37168)     ( 0.34907,-0.37168)
    1     ( 0.54890,-0.35669)     ( 0.54890,-0.35669)
    2     ( 0.74776,-0.31175)     ( 0.74776,-0.31175)
    3     ( 0.94459,-0.23702)     ( 0.94459,-0.23702)
    4     ( 1.13850,-0.13274)     ( 1.13850,-0.13274)
    5     ( 1.32850, 0.00074)     ( 1.32850, 0.00074)
    6     ( 1.51370, 0.16298)     ( 1.51370, 0.16298)
```